
Appendix A
An Introduction to the Event-B Modelling
Method

Thai Son Hoang

Abstract This appendix is a short introduction to the Event-B modelling method for
discrete transition systems. Important mechanisms for the step-wise development
of formal models, such as context extension and machine refinement, are discussed.
Consistency of the models is presented in terms of proof obligations and illustrated
with concrete examples.

A.1 Introduction

Event-B [2] is a modelling method for formalising and developing systems whose
components can be modelled as discrete transition systems. An evolution of the
(classical) B-method [1], Event-B is now centred around the general notion of
events, which are also found in other formal methods such as Action Systems [4],
TLA [6] and UNITY [5].

Event-B models are organised in terms of two basic constructs: contexts and
machines. Contexts specify the static part of a model, whereas machines specify
the dynamic part. The role of the contexts is to isolate the parameters of a formal
model and their properties, which are assumed to hold for all instances. A machine
encapsulates a transition system with the state specified by a set of variables and
transitions modelled by a set of guarded events.

Event-B allows models to be developed gradually via mechanisms such as con-
text extension and machine refinement. These techniques enable users to develop
target systems from their abstract specifications, and subsequently introduce more
implementation details. More importantly, properties that are proved at the abstract
level are maintained through refinement, and hence are also guaranteed to be satis-
fied by later refinements. As a result, correctness proofs of systems are broken down
and distributed amongst different levels of abstraction, which is easier to manage.

The rest of this appendix is structured as follows. We give a brief overview of
the Event-B mathematical language in Sect. A.2. In Sect. A.3, we give an informal
description of our running example. Subsequently, we show the basic constructs of

Table A.1 Definitions

Construct Definition

r ∈ S ↔ T r ∈ P(S × T)

f ∈ S �→ T f ∈ S ↔ T ∧ (∀x, y1, y2 ·x �→ y1 ∈ f ∧ x �→ y2 ∈ f ⇒ y1 = y2)

f ∈ S → T f ∈ S �→ T ∧ (∀x·x ∈ S ⇒ (∃y·x �→ y ∈ f ))

Event-B in Sects. A.4 (contexts) and A.5 (machines). We present the mechanisms
for context extension in Sect. A.6 and machine refinement in Sect. A.7.

A.2 The Event-B Mathematical Language

The basis for the formal models in Event-B is first-order logic and a typed set theory.
We are not going to give full details of the Event-B logic here. For more informa-
tion, we refer the reader to [2, 8]. We present several main elements of the Event-B
mathematical language that are important for understanding the Event-B models of
the example below.

The first-order logic of Event-B contains standard operators such as conjunc-
tion (∧), disjunction (∨), implication (⇒), negation (¬), equivalence (⇔), univer-
sal quantification (∀), and existential quantification (∃). Two constants are defined,
namely � and ⊥, denoting truth and falsity, respectively.

A.2.1 Set Theory

An important part of the mathematical language is its set-theoretical notation, with
the introduction of the membership predicate E ∈ S, meaning that expression E
is a member of set S. A set expression can be a variable (depending on its type).
Moreover, a set can be explicitly defined by listing its members (set extension), e.g.
{E1, . . . ,En}. Other basic set constructs include Cartesian product, power set, and
set comprehension. Given two set expressions S and T , the Cartesian product of S
and T , denoted S × T , is the set of mappings (ordered pairs) x �→ y, where x ∈ S and
y ∈ T . The power set of S, denoted P(S), is the set of all subsets of S. Finally, given
a list of variables x, a predicate P constraining x and an expression E depending on
x, the set comprehension {x·P | E} is the set of elements E where P holds.

A key feature of the Event-B set-theoretical notation is the models of relations
as sets of mappings. Different types of relations and functions are also defined as
sets of mappings with different additional properties. Given two set expressions S
and T , S ↔ T denotes the set of all binary relations from S to T . Similarly, S �→ T
denotes the set of all partial functions from S to T , and S → T denotes the set of all
total functions from S to T . Definitions of these relations can be seen in Table A.1,
expressed using set memberships.

Table A.2 Calculating
well-definedness conditions
using L

Formula Well-definedness condition

x �
¬P L (P )

P ∧ Q L (P ) ∧ (P ⇒ L (Q))

∀x ·P ∀x ·L (P )

E1 ÷ E2 L (E1) ∧ L (E2) ∧ E2 �= 0

E1 ≤ E2 L (E1) ∧ L (E2)

card(E) L (E) ∧ finite(E)

f (E) L (E) ∧ f ∈ S �→ T ∧ E ∈ dom(f )

Intuitively, a binary relation r from S to T is a set of mappings x �→ y, where
x ∈ S and y ∈ T . A partial function f from S to T is a binary relation from S to T ,
where each element x in S has at most one mapping to T . A total function f from
S to T is a partial function from S to T , where each element x in S has exactly one
mapping to T .

A.2.2 Types

Variables in Event-B are strongly typed. A type in Event-B can be built-in (e.g.,
BOOL, Z) or user-defined. Moreover, given types T , T1 and T2, the Cartesian prod-
uct T1 × T2 and the power set P(T) are also types. In contrast with most strongly
typed programming languages, the types of variables in Event-B are not presented
when they are declared. Instead, they are inferred from constraining properties of
variables. Typically, a property of the form x ∈ E, where E is of type P(T), allows
us to infer that x has type T .

A.2.3 Well-Definedness

Event-B requires every formula to be well defined [7, 8]. Informally, one has to prove
that partial functions (either predefined, e.g. division ÷, or user-defined) are never
evaluated outside of their domain. Ill-defined expressions (such as x ÷ 0) should
be avoided. A syntactic operator L is used to map formulae to their correspond-
ing well-definedness conditions. Table A.2 shows the definition of well-definedness
condition using L for formulae in Event-B. Here x is a variable, P and Q are pred-
icates, E, E1, E2 are expressions, and f is a binary relation from S to T . Moreover,
dom(f ) denotes the domain of f , i.e. the set of all elements in S that connect to an
element in T .

Notice that by using L , we assume a well-definedness order from left to right
(e.g., for P ∧ Q) for formulae (this is similar to evaluating conditional statements,
e.g. && or ||, in several programming languages).

A.2.4 Sequents

Event-B defines proof obligations, which must be proved to show that formal mod-
els fulfil their specified properties. Often these verification conditions are expressed
in terms of sequents. A sequent H � G means that the goal G holds under the
assumption of the set of hypotheses H . The obligations are discharged using cer-
tain inference rules, which we will not describe here. Instead, we will give informal
justification of how proof obligations can be discharged. The purpose of presenting
proof obligations within this appendix is to illustrate various conditions that need to
be proved to maintain the consistency of the example.

A.3 Example. A Course Management System

The running example that we are going to use for illustrating Event-B is a course
management system. We describe a requirements document of the system as fol-
lows. A club has some fixed members; amongst them are instructors and partici-
pants. Note that a member can be both an instructor and a participant.

ASM 1 Instructors are members of the club.

ASM 2 Participants are members of the club.

There are predefined courses that can be offered by a club. Each course is asso-
ciated with exactly one fixed instructor.

ASM 3 There are predefined courses.

ASM 4 Each course is assigned to one fixed instructor.

A course is either opened or closed and is managed by the system.

REQ 5 A course is either opened or closed.

REQ 6 The system allows a closed course to be opened.

REQ 7 The system allows an opened course to be closed.

The number of opened courses is limited.

REQ 8
The number of opened courses

cannot exceed a given limit.

Only when a course is opened, can participants register for the course. An impor-
tant constraint for registration is that an instructor cannot attend his own courses.

REQ 9 Participants can only register for an opened course.

REQ 10 Instructors cannot attend their own courses.

In subsequent sections, we develop a formal model based on the above require-
ments document. In particular, we will refer to the above requirements in order to
justify how they are formalised in the Event-B model.

A.4 Contexts

A context may contain carrier sets, constants, axioms, and theorems. Carrier sets are
user-defined types. By convention, a carrier set s is non-empty, i.e., satisfying s �= ∅,
and maximal, i.e., satisfying ∀x·x ∈ s. Constants c denote static objects within the
development.1 Axioms are presumed properties of carrier sets and constants. Theo-
rems are derived properties of carrier sets and constants. Carrier sets and constants
model the parameters of development. Moreover, axioms state parameter proper-
ties, assumed to hold for all their possible instances. A context C with carrier sets s,
constants c, axioms A(s, c), and theorems T(s, c) can be presented as follows.

1When referring to carrier sets s and constants c, we usually allow for multiple carrier sets and
constants, i.e., they may be “vectors”.

carrier sets: s constants: c

axioms:
axm : A(s, c)
thm : T(s, c)

Note that we present axioms and theorems using different labels, i.e., axm and
thm. Later on, we also use different labels for other modelling elements, such as
invariants (inv), guards (grd) and actions (act).

Proof obligations are generated to ensure that the theorems are derivable from
previously defined axioms. This is captured by the following proof obligation rule,
called THM:

A(s, c) � T(s, c). (THM)

A.4.1 Example. Context coursesCtx

In this initial model, we focus on the opening and closing of courses by the system.
As a result, our initial context coursesCtx contains a carrier set CRS denoting the set
of courses that can be offered by the club (ASM 3). Moreover, coursesCtx includes
a constant m denoting the maximum number of courses that the club can have at
the same time (with respect to requirement REQ 8). The context coursesCtx is as
follows.

carrier sets: CRS constants: m

axioms:
axm0_1 : finite(CRS)

axm0_2 : m ∈ N1
thm0_1 : 0 < m
axm0_3 : m ≤ card(CRS)

Note that we label the axioms and theorems with prefixes denoting the role of the
modelling elements, i.e., axm and thm, with some numbers. For example, axm0_1
denotes the first (i.e., 1) axiom for the initial model (i.e., 0). We apply this systematic
labelling throughout our development.

The assumptions on CRS and m are captured by the axioms and theorems as
follows. Axiom axm0_1 states that CRS is finite. Axiom axm0_2 states that m is a
member of the set of natural numbers (i.e., m is a natural number). Finally, axiom
axm0_3 states that m cannot exceed the number of possible courses that can be
offered by the club, represented as card(CRS), the cardinality of CRS. A derived
property of m is presented as theorem thm0_1.

thm0_1/THM A proof obligation is generated for thm0_1 as follows. Notice
that axm0_3 does not appear in the set of hypotheses for the obligation, since it is
declared after thm0_1. By convention, each proof obligation is labelled according to
the element involved and the name of the proof obligation rule. Here thm0_1/THM
indicates that it is a THM proof obligation for thm0_1.

finite(CRS)

m ∈ N1
�

0 < m

thm0_1/THM

The obligations can be trivially discharged since N1 is the set of all positive natural
numbers, i.e. {1,2, . . .}.

axm0_3/WD It is required to prove that axm0_3 is well defined. The correspond-
ing proof obligation is as follows.

finite(CRS)

m ∈ N

0 ≤ m
�

finite(CRS)

thm0_1/WD

Since the goal appears amongst the hypotheses, the proof obligation can be dis-
charged trivially. Note that the order of appearance of the axioms is important. In
particular, axm0_1 needs to be declared before axm0_3.

A.5 Machines

Machines specify behavioural properties of Event-B models. In order to have access
to information on context C, defined in Sect. A.4, machine M must connect with C.
When machine M sees context C, it has access to C’s carrier sets s and constants c,
to refer to them when modelling, and C’s axioms A(s, c) and theorems T(s, c), to
use them as assumptions during proving. For clarity, in the following presentation
we will not refer explicitly to the modelling elements of C. Note that in general a
machine can see several contexts.

Machines M may contain variables, invariants, theorems, events, and a variant.
Variables v define the state of a machine and are constrained by invariants I(v).
Theorems R(v) are additional properties of v derivable from I(v).

variables: v

invariants:
inv : I(v)
thm : R(v)

A proof obligation (also called THM) is generated to prove that the theorem R(v) is
derivable from I(v):

I(v) � R(v). (THM)

Possible state changes are described by events (see Sect. A.5.1). The variant is
used to prove convergence properties of events (see Sect. A.5.2).

A.5.1 Events

An event e can be represented by the term

e =̂ any x where G(x, v) then Q(x, v) end , (A.1)

where x stands for the event parameters,2 G(x, v) is the guard (the conjunction of
one or more predicates) and Q(x, v) is the action. The guard states the necessary
condition under which an event may occur. The event is said to be enabled in a
state, if there exists some value for its parameter x that makes its guard G(x, v) hold
in this state. The action describes how the state variables evolve when the event
occurs. We use the short form

e =̂ when G(v) then Q(v) end (A.2)

when the event does not have any parameters, and we write

e =̂ begin Q(v) end (A.3)

when, in addition, the event guard always holds (i.e., equals �). A dedicated event
in the form of (A.3) is used for the initialisation event (usually represented by init).
Note that events may be annotated with their convergence status, which we will look
at in Sect. A.5.2.

The action of an event is composed of one or more assignments of the form

a := E(x, v) (A.4)

or

a :∈ E(x, v) (A.5)

or

a :| P
(

x, v, a′), (A.6)

where a is some of the variables contained in v, E(x, v) is an expression, and
P(x, v, a′) is a predicate. Note that the variables on the left-hand side of the assign-
ments contained in an action must be disjoint. In (A.5), a must be a single variable,
whereas it can be a vector of variables in (A.4) and (A.6). In particular, in (A.4), if a

is a vector containing n > 0 variables, then E must also be a vector of expressions,
one for each of the n variables. Assignments of the form (A.4) are deterministic,
whereas assignments of the other two forms are nondeterministic. In (A.5), a is
assigned an element of a set E(x, v). (A.6) refers to P, which is a before-after pred-
icate relating the values v (before the action) and a′ (afterwards). (A.6) is also the

2When referring to variables v and parameters x, we usually allow for multiple variables and pa-
rameters, i.e., they may be “vectors”.

most general form of assignment and nondeterministically selects an after-state a′
satisfying P and assigns it to a. Note that the before-after predicates for the other
two forms are as expected; namely, a′ = E(x, v) and a′ ∈ E(x, v), respectively.

All assignments of an action Q(x, v) occur simultaneously, which is expressed by
conjoining their before-after predicates. Hence each event corresponds to a before-
after predicate Q(x, v, v′) established by conjoining all before-after predicates asso-
ciated with each assignment and b = b′, where b is unchanged variables. Note that
the initialisation init therefore corresponds to an after predicate K(v′), since there
are no states before initialisation.

A.5.1.1 Proof Obligations

Below we describe some important proof obligation rules for Event-B machines,
namely, invariant establishment and preservation, and action feasibility.

Invariant Establishment and Preservation An essential feature of an Event-B
machine M is its invariant I(v). It shows properties that hold in every reachable state
of the machine. Obviously, this does not hold a priori for any machines and invari-
ants, and therefore must be proved. A common technique for proving an invariant
property is to prove it by induction: (1) to prove that the property is established by
the initialisation init (invariant establishment), and (2) to prove that the property is
maintained whenever variables change their values (invariant preservation).

Invariant establishment states that any possible state after initialisation given by
the after predicate K(v′) must satisfy the invariant I. The proof obligation rule is as
follows:

K
(

v′) � I
(

v′). (INV)

Invariant preservation makes it necessary to prove that every event occurrence
reestablishes the invariants I. More precisely, for every event e, assuming the invari-
ants I and e’s guard G, we must prove that the invariants still hold in any possible
state after the event execution given by the before-after predicate Q(x, v, v′). The
proof obligation rule is as follows:

I(v),G(x, v),Q
(

x, v, v′) � I
(

v′). (INV)

Note that in practice, by the property of conjunctivity, we can prove the estab-
lishment and preservation of each invariant separately.

Feasibility Feasibility states that the action of an event is always feasible when-
ever the event is enabled. In other words, there are always possible after values
for the variables satisfying the before-after predicate. In practice, we prove feasi-
bility for individual assignment of the event action. For deterministic assignments,
feasibility holds trivially. The feasibility proof obligation generated for a nondeter-
ministic assignment of the form a :| P(x, v, a′) is as follows:

I(v),G(x, v) � ∃a′ ·P(

x, v, a′) . (FIS)

A.5.2 Event Convergence

A set of events can be proved to collectively converge. In other words, these events
cannot take control forever and hence one of the other events eventually occurs. We
call these events convergent. To prove this, one specifies a variant V which maps
a state v to a natural number. One then proves that each convergent event strictly
decreases V . More precisely, let e be a convergent event where v is the state before
executing e and v′ is the state after. Then for each such e, v, and v′, one proves
that V(v′) < V(v), additionally assuming all invariants and the e guard. Since the
variant maps a state to a natural number, V induces a well-founded ordering on
system states given by the strictly less than order (<) of their images under V . The
following proof obligation rules apply to every convergent event where VAR ensures
the decrement of the variant and NAT ensures that the variant is a natural number
when the event is enabled:

I(v),G(x, v),Q
(

x, v, v′) � V
(

v′) < V(v) (VAR)

I(v),G(x, v) � V(v) ∈ N. (NAT)

Note that in some cases the convergence of some events can be shown only in a
later refinement, but not immediately. In this case, their convergence is anticipated
and we must prove that V(v′) ≤ V(v), that is, these anticipated events do not increase
the variant. Anticipated events must obey NAT and the following proof obligation
rule, also called VAR:

I(v),G(x, v),Q
(

x, v, v′) � V
(

v′) ≤ V(v). (VAR)

As mentioned above, variant V is a natural number. Alternatively, V can be a
finite set expression. In this case, for convergent events, one has to prove that it
decreases the variant according to the strict subset inclusion ⊂ ordering. For antici-
pated events, we ensure that these events do not increase the variant by proving that
V(v′) ⊆ V(v). The proof obligation rule VAR is adapted accordingly.

For proving that the variant V is a finite set, the following proof obligation rule
called FIN applies:

I(v) � finite
(

V(v)
)

. (FIN)

Note that FIN needs to be proved once, i.e., it does not depend on the set of
convergent and anticipated events (cf. NAT).

The convergence attribute of an event is denoted by the keyword status with
three possible values: convergent, anticipated, and ordinary (for events which are
neither convergent nor anticipated). Events are ordinary by default.

A.5.3 Deadlock-Freeness

A machine M is said to be deadlocked in some state if all of its events are disabled
in that state. Deadlock-freeness for M ensures that there are always some enabled

events during the execution of M. Assume that M contains a set of n events ei (i ∈
1, . . . , n) of the following form:

ei =̂ any xi where Gi (xi , v) then Qi (xi , v) . end

The proof obligation rule for deadlock-freeness3 is as follows:

I(v) � (∃x1 ·G1(x1, v)
) ∨ · · · ∨ (∃xn ·Gn(xn, v)

)

. (DLF)

A.5.4 Example. Machine m0

We develop machine m0 of the initial model, focusing on courses opening and clos-
ing. This machine sees context coursesCtx as developed in Sect. A.4.1, and as a
result has access to the carrier set CRS and constant m. We model the set of opened
courses by a variable, namely crs (REQ 5). Invariant inv0_1 states that it is a subset
of available courses CRS. A consequence of this invariant and of axiom axm0_1 is
that crs is finite, and this is stated in m0 as theorem thm0_2. Requirement REQ 8 is
directly captured by invariant inv0_2: the number of opened courses, i.e., card(crs)
is bounded above by m. Initially, all courses are closed; hence crs is set to the empty
set (∅).

variables: crs

invariants:
inv0_1 : crs ⊆ CRS
thm0_2 : finite(crs)
inv0_2 : card(crs) ≤ m

init
begin

crs := ∅

end

We model the opening and closing of courses using two events OpenCourses and
CloseCourses as follows (REQ 6 and REQ 7).

OpenCourses
status ordinary
when

grd0_1 : card(crs) �= m
thm0_3 : crs �= CRS

then
act0_1 : crs :| crs ⊂ crs′ ∧ card(crs′) ≤ m

end

CloseCourses
status anticipated
any cs where

grd0_1 : cs ⊆ crs
grd0_2 : cs �= ∅

then
act0_1 : crs := crs \ cs

end

We deliberately choose to model these events using different features of Event-B.
In OpenCourses, we use a nondeterministic action to model the fact that some new
courses are opened, i.e. crs ⊂ crs′, as long as the number of opened courses will not
exceed its limit, i.e. card(crs′) ≤ m. The guard of the event states that the current
number of opened courses has not yet reached the limit.

3Typically, this is encoded as a theorem in the machine after all invariants.

CloseCourses models the set of courses that are going to be closed using pa-
rameter cs. It is a non-empty set of currently opened courses which is captured by
CloseCourses’ guard. The action is modelled in a straightforward way by removing
cs from set crs.

We set the convergence status for OpenCourses and CloseCourses to be ordinary
and anticipated, respectively. We postpone the reasoning about the convergence of
CloseCourses till later refinements. Our intention is to prove that there can only
be a finite number of occurrences of CloseCourses between any two OpenCourses
events.

A.5.4.1 Proof Obligations

We present some of the obligations to illustrate what needs to be proved for the con-
sistency of m0. We applied the proof obligation rules as shown earlier in this section.
Notice that we can take the axioms and theorems of the seen context coursesCtx as
hypotheses in the proof obligations. For clarity, we show only the parts of the hy-
potheses that are relevant for discharging the proof obligations. Moreover, we also
show the proof obligations in their simplified forms, e.g. when event assignments
are deterministic.

thm0_2/THM This obligation corresponds to the rule THM, in order to ensure
that thm0_2 is derivable from previously declared invariants.

. . .

finite(CRS)

crs ⊆ CRS
�

finite(crs)

thm0_2/THM

The proof obligation holds trivially since crs is a subset of a finite set, i.e., CRS.

init/inv0_2/INV This obligation ensures that the initialisation init establishes in-
variant inv0_2.

. . .

0 ≤ m
�

card(∅) ≤ m

init/inv0_2/INV

Since the cardinality of the empty set ∅ is 0, the proof obligation holds trivially.

OpenCourses/thm0_3/THM This obligation ensures that thm0_3 is derivable
from the invariants and the previously declared guards of OpenCourses.

. . .

m ≤ card(CRS)

crs ∈ P(CRS)

card(crs) ≤ m
card(crs) �= m

�
crs �= CRS

OpenCourses/thm0_3/THM

Informally, we can derive from the hypotheses that card(crs) < card(CRS); hence
crs must be different from CRS.

OpenCourses/act0_1/FIS This obligation corresponds to rule FIS and ensures
that the nondeterministic assignment of OpenCourses is feasible when the event is
enabled.

. . .

crs �= CRS
card(crs) ≤ m
card(crs) �= m

�
∃crs′ ·crs ⊂ crs′ ∧ card(crs′) ≤ m

OpenCourses/act0_1/FIS

The reasoning about the proof obligation is as follows. Since crs is different
from CRS, there exists an element c which is closed, i.e., not in crs. By adding c to
the set of opened courses, we strictly increase the number of opened courses by 1.
Moreover, the number of opened courses after executing the event is still within the
limit since originally it is strictly below the limit.

CloseCourses/inv0_2/INV This obligation corresponds to rule INV and is sim-
plified accordingly since the assignment is deterministic. The purpose of the obliga-
tion is to prove that CloseCourses maintains invariant inv0_2.

. . .

card(crs) ≤ m
�

card(crs \ cs) ≤ m

CloseCourses/inv0_2/INV

Since removing some courses cs from the set of opened courses crs can only reduce
its number, the proof obligation can be trivially discharged.

DLF/THM The deadlock-freeness condition is encoded as theorem DLF of ma-
chine m0, which results in the following proof obligation.

. . .

0 < m
�

(card(crs) �= m) ∨ (∃cs·cs ⊆ crs ∧ cs �= ∅)

DLF/THM

We reason as follows. If card(crs) �= m, the goal trivially holds. Otherwise, if
card(crs) = m, since m �= 0, crs �= ∅. As a result, we can prove that ∃cs·cs ⊆
crs ∧ cs �= ∅ by instantiating cs with crs.

A.6 Context Extension

Context extension is a mechanism for introducing more static details into an Event-B
development. A context can extend one or more contexts. When describing a context
D as extending another context C, we call C and D the abstract and concrete context,
respectively. By extending C, D “inherits” all the abstract elements of C, i.e., carrier
sets, constants, axioms and theorems. This means that (1) a context extending D
also implicitly extends C, and (2) a machine seeing D also implicitly sees C. As a
result, proof obligation rule THM for D also has additional assumptions in the form
of axioms and theorems from the abstract context C.

Subsequently, we present three new contexts that we use in the next refinement
of our running example.

A.6.1 Context membersCtx

This is an initial context (i.e., it does not extend any other context) containing a
carrier set MEM. MEM represents the set of club members, with an axiom stating
that it is finite.

carrier sets: MEM
axioms:

axm1_1 : finite(MEM)

A.6.2 Context participantsCtx

This context extends the previously defined context membersCtx and is as follows.

constants: PRTCPT

axioms:
axm1_2 : PRTCPT ⊆ MEM
thm1_1 : finite(PRTCPT)

Constant PRTCPT denotes the set of participants that must be members of the club
as specified in ASM 2 (axm1_2). Theorem thm1_1 states that there can be only a
finite number of participants, which gives rise to the following trivial proof obliga-
tion.

finite(MEM)

PRTCPT ⊆ MEM
�

finite(PRTCPT)

thm1_1/THM

An important point is that axiom axm1_1 of the abstract context membersCtx ap-
pears as a hypothesis in the proof obligation.

A.6.3 Context instructorsCtx

This context extends two contexts, coursesCtx and membersCtx, and introduces
two constants, namely INSTR and instrs. INSTR models the set of instructors that
are members of the club as specified by ASM 1 (axm1_3). Constant instrs models
the relationship between courses and instructors and is constrained by axm1_4: it
is a total function from CRS to INSTR, and hence directly formalises requirement
ASM 4. Recall the definition of total function f from set S to set T : f is a relation
from S to T where every element in S has exactly one mapping to some element
in T .

constants: INSTR, instrs

axioms:
axm1_3 : INSTR ⊆ MEM
axm1_4 : instrs ∈ CRS → INSTR

The hierarchy of context extensions for our example is summarised in the fol-
lowing diagram.

A.7 Machine Refinement

Machine refinement is a mechanism for introducing details about the dynamic prop-
erties of a model [2]. For more details on the theory of refinement, we refer the
reader to the Action System formalism [4], which has inspired the development of
Event-B. We present here the proof obligations defined in Event-B, related to refine-
ment. When speaking about machine N refining another machine M, we refer to M
as the abstract machine and to N as the concrete machine.

Despite the fact that the formal definition of Event-B refinement does not distin-
guish between superposition refinement and data refinement, we illustrate them in
separate sections to show different aspects of the two. In superposition refinement,
the abstract variables of M are retained in the concrete machine N, with possibly
some additional concrete variables. In data refinement, the abstract variables v are

replaced by concrete variables w and, subsequently, the connections between M and
N are represented by the relationship between v and w. In fact, more often, Event-
B refinement is a mixture of both superposition and data refinement: some of the
abstract variables are retained, while others are replaced by new concrete variables.

A.7.1 Superposition Refinement

As mentioned earlier, in superposition refinement, variables v of the abstract ma-
chine M are kept in the refinement, i.e. as part of the state of N. Moreover, N can
have some additional variables w. The concrete invariants J(v,w) specify the rela-
tionship between the old and new variables. Each abstract event e is refined by a
concrete event f (later on we will relax this one-to-one constraint). Assume that the
abstract event e and the concrete event f are as follows:

e =̂ any x where G(x, v) then Q(x, v) end

f =̂ any x where H(x, v,w) then R(x, v,w) end

We assume now that e and f have the same parameters x. The more general case,
where the parameters are different, is presented in Sect. A.7.2.

Somewhat simplifying, we say that f refines e if the guard of f is stronger than that
of e (guard strengthening), concrete invariants J are maintained by f, and abstract
action Q simulates the concrete action R (simulation). These conditions are stated
as the following proof obligation rules:

I(v), J(v,w),H(x, v,w) � G(x, v) (GRD)

I(v), J(v,w),H(x, v,w),R
(

x, v,w, v′,w′) � Q
(

x, v, v′) (SIM)

I(v), J(v,w),H(x, v,w),R
(

x, v,w, v′,w′) � J
(

v′,w′) (INV)

In particular, if the guard and action of an abstract event are retained in the concrete
event, the proof obligations GRD and SIM are trivial; hence we only need to consider
INV for proving that the gluing invariants are reestablished.

Proof obligations are generated to ensure that each assignment of concrete event
f is feasible. In the case where the action of the abstract event is retained in f, we
only need to prove the feasibility of any additional assignment.

In the course of refinement, new events are often introduced into a model. New
events must be shown to refine the implicit abstract event SKIP, which does nothing,
i.e., does not modify abstract variables v.

A.7.1.1 Machine m1

Machine m1 sees contexts instructorsCtx and participantsCtx. As a result, it implic-
itly sees coursesCtx and membersCtx. Variable crs is retained in this refinement.

An additional variable prtcpts representing information about course participants is
introduced. Invariant inv1_1 models prtcpts as a relation between the set of opened
courses crs and the set of participants PRTCPT . Requirement REQ 10 is directly
modelled by invariant inv1_2: for every opened course c, the instructor of this
course, i.e., instrs(c), is not amongst its participants, represented by prtcpts[{c}].

variables: crs,prtcpts

invariants:
inv1_1 : prtcpts ∈ crs ↔ PRTCPT
inv1_2 : ∀c · c ∈ crs ⇒ instrs(c) /∈ prtcpts[{c}]

init
begin

. . .

prtcpts := ∅

end

Initially, there are no opened courses; hence prtcpts is assigned to be ∅.
The original abstract event OpenCourses stays unchanged in this refinement,

while an additional assignment is added to CloseCourses to update prtcpts by re-
moving the information about the set of closing courses cs from it.

CloseCourses
status anticipated
any cs where

. . .

then
. . .

act1_2 : prtcpts := cs �− prtcpts
end

A new event is added, namely Register, to model the registration of a participant
p for an opened course c. The guard of the event ensures that p is not the instructor
of the course (grd1_3) and is not yet registered for the course (grd1_4). The action
of the event updates prtcpts accordingly by adding the mapping c �→ p to it.

Register
status convergent
any p, c where

grd1_1 : p ∈ PRTCPT
grd1_2 : c ∈ crs
grd1_3 : p �= instrs(c)
grd1_4 : c �→ p /∈ prtcpts

then
act1_1 : prtcpts := prtcpts ∪ {c �→ p}

end

We attempt to prove that Register is convergent and CloseCourses is anticipated
using the following variant.

variant: (crs × PRTCPT) \ prtcpts

The variant is a set of mappings; each links an opened course to a participant
who has not registered for the respective course.

We present some of the important proof obligations for m1. Proof obligations
GRD and SIM are trivial for events OpenCourses and CloseCourses. Consequently,
we only need to consider INV for these old events.

CloseCourses/inv1_2/INV This obligation is to ensure that inv1_2 is main-
tained by CloseCourses. The obligation is trivial, in particular, because, given that
c /∈ cs, (cs �− prtcpts)[{c}] is the same as prtcpts[{c}].

. . .

∀c · c ∈ crs ⇒ instrs(c) /∈ prtcpts[{c}]
�

∀c · c ∈ crs \ cs ⇒ instrs(c) /∈ (cs �− prtcpts)[{c}]
CloseCourses/inv1_2/INV

Register/inv1_1/INV This obligation is to guarantee that inv1_1 is maintained
by the new event Register.

. . .

prtcpts ∈ crs ↔ PRTCPT
p ∈ PRTCPT
c ∈ crs

�
prtcpts ∪ {c �→ p} ∈ crs ↔ PRTCPT

Register/inv1_1/INV

FIN This obligation is to ensure that the declared variant used for proving conver-
gence of events is finite (FIN). This is trivial, since the set of opened courses crs and
the set of participants PRTCPT are both finite.

. . .

finite(crs)
finite(PRTCPT)

�
finite((crs × PRTCPT) \ prtcpts)

FIN

CloseCourses/VAR This proof obligation corresponds to rule VAR, ensuring that
anticipated event CloseCourses does not increase the variant.

. . .

�
((crs \ cs) × PRTCPT) \ (cs �− prtcpts)

⊆
(crs × PRTCPT) \ prtcpts)

CloseCourses/VAR

Register/VAR This proof obligation corresponds to rule VAR, ensuring that the
convergent event Register decreases the variant. This is trivial since a new mapping
c �→ p is added to prtcpts, effectively increasing it, and hence strictly decreasing the
variant.

. . .

c �→ p /∈ prtcpts
�

(crs × PRTCPT) \ (prtcpts ∪ {c �→ p}))
⊂

(crs × PRTCPT) \ prtcpts)

Register/VAR

A.7.2 Data Refinement

In data refinement, abstract variables v are removed and replaced by concrete vari-
ables w. The states of abstract machine M are related to the states of concrete ma-
chine N by gluing invariants J(v,w). In Event-B, the gluing invariants J are declared
as invariants of N and also contain the local concrete invariants, i.e., those constrain-
ing only concrete variables w.

Again, we assume a one-to-one correspondence between an abstract event e and
a concrete event f. Let e and f be as follows:4

e =̂ any x where G(x, v) then Q(x, v) end

f =̂ any y where H(y,w) then R(y,w) end

As with superposition refinement, we can say that f refines e if the guard of f is
stronger than the guard of e (guard strengthening), and the gluing invariants J(v,w)

establish a simulation of f by e (simulation). This condition is captured by the fol-
lowing proof obligation rule:

I(v)
J(v,w)

H(y,w)

R
(

y,w,w′)

�
∃x, v′ ·G(x, v) ∧ Q

(

x, v, v′) ∧ J
(

v′,w′)

(A.7)

In order to simplify and split the above proof obligation, Event-B introduces the
notion of “witnesses” for the abstract parameters x and the after value of the ab-
stract variables v′. Witnesses are predicates of the form W1(x, y, v,w,w′) (for x)

4Concrete events may be annotated with abstract events name and witnesses, which we will show
later.

and W2(v′, y, v,w,w′) (for v′), which are required to be feasible. The corresponding
proof obligations are as follows:

I(v), J(v,w),H(y,w),R
(

y,w,w′) � ∃x·W1
(

x, y, v,w,w′) (WFIS)

I(v), J(v,w),H(y,w),R
(

y,w,w′) � ∃v′ ·W2
(

v′, y, v,w,w′) (WFIS)

Typically, witnesses are declared deterministically, i.e. in the form x = · · · or v′ =
· · · . In these cases, witnesses are trivially feasible; hence the corresponding proof
obligations are omitted.

Given the witnesses, the refinement proof obligation (A.7) is replaced by three
different proof obligations as follows:

I(v), J(v,w),H(y,w),W1
(

x, y, v,w,w′) � G(x, v) (GRD)

I(v), J(v,w),H(y,w),R
(

y,w,w′),W1
(

x, y, v,w,w′),W2
(

v′, y, v,w,w′) � Q
(

x, v, v′)

(SIM)

I(v), J(v,w),H(y,w),R
(

y,w,w′),W1
(

x, y, v,w,w′),W2
(

v′, y, v,w,w′) � J
(

v′,w′)

(INV)

The concrete event f can be denoted by the abstract event e (using keyword refines)
and the witnesses (using keyword with) as follows:

f
refines e
any y where

H(y,w)

with
x : W1(x, y, v,w,w′)
v′ : W2(v′, y, v,w,w′)

then
R(y,w)

end

The action of the concrete event is required to be feasible. The corresponding
proof obligation FIS is similar to the one presented for the abstract machine, with
the exception that both abstract and gluing invariants can be assumed.

For newly introduced events, as with superposition refinement, they must be
proved to refine the implicit abstract event SKIP, which is unguarded and does noth-
ing, i.e. does not modify abstract variables v. In this case, GRD is trivial, since the
abstract guard is �. For SIM and INV, we omit references to W1 (since there are no
parameters for the abstract SKIP event). Moreover, the witness W2 for v′ is trivial:
v′ = v.

As mentioned earlier, in general, Event-B refinement is a mixture of both super-
position and data refinement. Often, some (not all) abstract variables are retained

in the refinement, while the other abstract variables are replaced by new concrete
variables. Similarly, some abstract parameters can be present in the concrete event,
where other parameters are replaced by some new concrete ones. In general, we
only need to give witnesses to disappearing variables and parameters.

The one-to-one correspondence between the abstract and concrete events can be
relaxed. When an abstract event e is refined by more than one concrete event f, we
say that the abstract event e is split and prove that each concrete f is a valid refine-
ment of the abstract event. Conversely, several abstract events e can be refined by
one concrete f. We say that these abstract events are merged together. A requirement
for merging events is that the abstract events must have identical actions. When
merging events, we need to prove that the guard of the concrete event is stronger
than the disjunction of the guards of the abstract events.

The concrete machine N can be proved to be relatively deadlock-free with re-
spect to the abstract machine M. It means that if M can continue in some state, so
can N. Assume that M contains a set of n events ei (i ∈ 1, . . . , n) of the following
form:

ei =̂ any xi where Gi (xi , v) then Qi (xi , v) end

Assume that N contains a set of m events fj (j ∈ 1, . . . ,m) of the following
form:

fj =̂ any yi where Hi (yi ,w) then Ri (yi ,w) end

The proof obligation rule for relative deadlock-freeness5 is as follows:

I(v)
J(v,w)
(∃x1 ·G1(x1, v)

) ∨ · · · ∨ (∃xn ·Gn(xn, v)
)

�
(∃y1 ·H1(y1,w)

) ∨ · · · ∨ (∃ym ·Hm(ym,w)
)

(REL_DLF)

A.7.2.1 Machine m2

We perform a data refinement by replacing abstract variables crs and prtcpts by
a new concrete variable atnds. This machine does not explicitly model any re-
quirements from Sect. A.3: it implicitly inherits requirements from previous ab-
stract machines. As stated in invariant inv2_1, atnds is a partial function from
CRS to some set of participants (i.e., member of P(PRTCPT)). Invariants inv2_2
and inv2_3 act as gluing invariants, linking abstract variables crs and prtcpts
with concrete variable atnds. Invariant inv2_2 specifies that crs is the domain
atnds. Invariant inv2_3 states that for every opened course c, the set of partic-

5Typically, this is encoded as a theorem in N after declaration of all invariants.

ipants attending that course represented abstractly as prtcpts[{c}] is the same as
atnds(c).

variables: atnds

invariants:
inv2_1 : atnds ∈ CRS �→ P(PRTCPT)

inv2_2 : crs = dom(atnds)
inv2_3 : ∀c·c ∈ crs ⇒ prtcpts[{c}] = atnds(c)
thm2_1 : finite(atnds)

init
begin

atnds := ∅

end

We illustrate our data refinement by the following example. Assume that the avail-
able courses CRS are {c1, c2, c3}, with c1 and c2 being opened, i.e., crs = {c1, c2}.
Assume that c1 has no participants, and p1 and p2 are attending c2. Abstract vari-
able prtcpts hence contains two mappings as follows: {c2 �→ p1, c2 �→ p2}. The
same information can be represented by the concrete variable atnds as follows:
{c1 �→∅, c2 �→ {p1,p2}}.

We refine the events using data refinement as follows. Event OpenCourses is
refined by OpenCourse, where one course (instead of possibly several courses) is
opened at a time. The course that is opening is represented by the concrete parame-
ter c.

OpenCourses
when

grd0_1 : card(crs) �= m
thm0_1 : crs �= CRS

then
act0_1 : crs :| crs ⊂ crs′ ∧ card(crs′) ≤ m

end

OpenCourse
refines OpenCourses
any c where

grd2_1 : c /∈ dom(atnds)
grd2_2 : card(atnds) �= m

with
crs′ = crs ∪ {c}

then
act2_1 : atnds(c) := ∅

end

The concrete guards ensure that c is a closed course and the number of opened
courses (card(atnds)) has not reached the limit m. The action of OpenCourse sets
the initial participants for the newly opened course c to be the empty set. In order
to prove the refinement relationship between OpenCourse and OpenCourses, we
need to give the witness for the after value of the disappearing variable crs′. In this
case, it is specified as crs′ = crs ∪ {c}, by adding the newly opened course c to the
original set of opened courses crs.

Abstract event CloseCourses is refined by concrete event CloseCourse, where
one course c (instead of possibly several courses cs) is closed at a time. The guard
and action of concrete event CloseCourse are as expected.

CloseCourses
status anticipated
any cs where

grd0_1 : cs ⊆ crs
grd0_2 : cs �= ∅

then
act0_1 : crs := crs \ cs
act2 : prtcpts := cs �− prtcpts

end

CloseCourse
refines CloseCourses
status convergent
any c where

grd2_1 : c ∈ dom(atnds)
with

cs = {c}
then

act2_1 : atnds := {c}�− atnds
end

We need to give the witness for the disappearing abstract parameter cs. It is specified
straightforwardly as cs = {c}. Notice also that we change the convergence status of
CloseCourse from anticipated to convergent. We use the following variant to prove
that CloseCourse is convergent.

variant: card(atnds)

The variant represents the number of mappings in atnds, and since it is a par-
tial function, it is also the same as the number of elements in its domain, i.e.
card(atnds) = card(dom(atnds)). As a result, the variant represents the number of
opened courses.

Event Register is refined as follows:6 references to crs and prtcpts in guard and
action are replaced by references to atnds.

(abs_)Register
any p, c where

grd1_1 : p ∈ PRTCPT
grd1_2 : c ∈ crs
grd1_3 : p �= instrs(c)
grd1_4 : c �→ p /∈ prtcpts

then
act1_1 : prtcpts := prtcpts ∪ {c �→ p}

end

(cnc_)Register
refines Register
any p, c where

grd2_1 : p ∈ PRTCPT
grd2_2 : c ∈ dom(attendees)

grd2_3 : p �= instrs(c)
grd2_4 : p /∈ atnds(c)

then
act2_1 : atnds(c) := atnds(c) ∪ {p}

end

We now show some proof obligations for proving the refinement of m1 by m2.

OpenCourse/act0_1/SIM This proof obligation corresponds to rule SIM, ensur-
ing that the action act0_1 of abstract event OpenCourses can simulate the action of
concrete event OpenCourse. Notice the use of the witness for crs′ as a hypothesis
in the obligation.

6We use prefixes (abs_) and (cnc_) to denote the abstract and concrete versions of the event,
respectively.

. . .

atnds ∈ CRS �→ P(PRTCPT)

crs = dom(attendees)
c /∈ dom(attendees)
card(attendees) �= m
crs′ = crs ∪ {c}

�
crs ⊂ crs′ ∧ card(crs′) ≤ m

OpenCourse/act0_1/SIM

CloseCourse/grd0_1/GRD This proof obligation corresponds to rule GRD, en-
suring that the guard of concrete event CloseCourse is stronger than the abstract
guard grd0_1 of abstract event CloseCourses. Note the use of the witness for cs as
a hypothesis in the obligation.

. . .

crs = dom(atnds)
c ∈ dom(atnds)
cs = {c}

�
cs ⊆ crs

CloseCourse/grd0_1/GRD

CloseCourse/NAT This proof obligation corresponds to rule NAT on page 220;
it ensures that the variant is a natural number when CloseCourse is enabled.

. . .

finite(atnds)
�

card(atnds) ∈ N

CloseCourse/NAT

CloseCourse/VAR This proof obligation corresponds to rule VAR on page 220;
it ensures that the variant is strictly decreased by CloseCourse. The obligation is
trivial since the variant represents the number of opened courses and CloseCourse
closes one of them.

. . .

atnds ∈ CRS �→ PRTCPT
c ∈ dom(atnds)

�
card({c}�− atnds) < card(atnds)

CloseCourse/VAR

A.8 Summary of the Development

The summary of the hierarchy of the development is illustrated in Fig. A.1.
Table A.3 summarises how assumptions and requirements are taken into account

in our formal development. Note that the last refinement, m2, does not explicitly

Fig. A.1 Development hierarchy

Table A.3 Requirements
Tracing Requirement Models Requirement Models

ASM 1 instructorsCtx REQ 6 m0

ASM 2 participantsCtx REQ 7 m0

ASM 3 coursesCtx REQ 8 m0

ASM 4 instructorsCtx REQ 9 m1

REQ 5 m0 REQ 10 m1

take into account any requirements. Indeed, the requirements are implicitly inherited
from the abstract machines through the refinement relationship.

The development is formalised and proved using the supporting Rodin platform7

for Event-B [3] and is available online.8 The summary of the proof statistics for the

Table A.4 Proof statistics

Constructs Proof obligations Automatic (%) Manual (%)

coursesCtx 2 2 (100 %) 0 (0 %)

membersCtx 0 0 (N/A) 0 (N/A)

instructorsCtx 0 0 (N/A) 0 (N/A)

participantsCtx 1 1 (100 %) 0 (0 %)

m0 11 8 (73 %) 3 (27 %)

m1 14 13 (93 %) 1 (7 %)

m2 29 26 (90 %) 3 (10 %)

Total 57 50 (88 %) 7 (12 %)

7At the time of writing, we use Rodin version 2.4.0.
8http://deploy-eprints.ecs.soton.ac.uk/371/.

http://deploy-eprints.ecs.soton.ac.uk/371/

development is shown in Table A.4. Around 50 % of the proof obligations appear
in m2, where we perform data refinement. Typically, data refinement involves radi-
cal changes to developments, since when replacing abstract variables with concrete
variables, it is also necessary to adapt the events accordingly.

